KIT de sobrevivencia em CALCULO

Prof. Doherty Andrade - DMA- UEM
doherty@gauss.dma.uem.br

<u>Máximos e Mínimos</u>

```
> with(linalg):
```

Warning, new definition for norm

Warning, new definition for trace

```
> with(plots):
```

Neste exemplos vamos usar o teste da derivada segunda para máximos, mínimos e sela para funções de varias variaveis.

Exemplo 1. considere a funcao: $\dfrac{3x^4 - 4x^3 - 12x^2 + 6}{12(1 + y^2)}$

```
> f:=(x,y) ->(3*x^4 -4*x^3 -12*x^2 +6)/(12*(1+y^2));
```

$$f := (x, y) \to \frac{3x^4 - 4x^3 - 12x^2 + 6}{12 + 12y^2}$$

Antes de trabalhar com fórmulas, pois temos uma fun ção de apenas duas variaveis podemos nos dar ao luxo de fazer o seu grafico que é uma superficie. O numerador é quartico

```
> plot(12*f(x,0), x=-2..3);
```



então nossa superficie é exatamente esta quartica com um "damping" como |y| crescendo (pois o denominador é menor). Isto é verificado se olharmos no grafico da superficie.

> **plot3d(f(x,y), x=-2..3, y=-4..4, style=patch, orientation=[-80,70],axes=normal, labels=[x,y,z], color=[-y, -y, 1]);**



Deste grafico vemos que f(x,y) tem três pontos criticos, todos no eixo x: uma sela perto de (-1, 0), um maximo local perto de (0, 0), e um minimo local perto de (2,0). O objetivo deste exemplo é obter o mesmo resultado usando fórmulas. Em situações envolvendo mais variáveis quando a visualiza ção é impossivel, as fórmulas são sempre convenientes.

Passo 1. Encontre os pontos criticos. Estes são por defini ção onde as derivadas parciais de primeira ordem de f(x,y) se anulam.

> **f1:= diff(f(x,y), x);**

$$f1 := \frac{12\,x^3 - 12\,x^2 - 24\,x}{12 + 12\,y^2}$$

> **f2:=diff(f(x,y), y);**

$$f2 := -24\,\frac{(3\,x^4 - 4\,x^3 - 12\,x^2 + 6)\,y}{(12 + 12\,y^2)^2}$$

Para encontrar os pontos critivos precisamos de resolver o sistema de equações f1=- e f2=0 (em geral não lineares) para valores de x e y. Vamos usar o Maple para resolver.

> **solve({f1=0, f2=0}, {x,y});**

$$\{y = 0, x = 0\}, \{y = 0, x = -1\}, \{y = 0, x = 2\}$$

Como alternativa ao cálculo das derivadas individualmente, podemos calcular o vetor gradiente:

> **grad(f(x,y), [x,y]);**

$$\left[\frac{12\,x^3 - 12\,x^2 - 24\,x}{12 + 12\,y^2}, -24\,\frac{(3\,x^4 - 4\,x^3 - 12\,x^2 + 6)\,y}{(12 + 12\,y^2)^2}\right]$$

Podemos agora encontrar os pontos onde o gradiente é zero. Para usar o "solve" primeiro convertemos para conjuntos:

> **convert(",set);**

e então resolver as equaçõoes

> **solve(",{x,y});**

$$\left\{-24\,\frac{(3x^4-4x^3-12x^2+6)\,y}{(12+12y^2)^2},\ \frac{12x^3-12x^2-24x}{12+12y^2}\right\}$$

$$\{y=0,x=0\},\{y=0,x=-1\},\{y=0,x=2\}$$

Usando as primeiras derivadas parciais explicitamente ou então via o gradiente o Maple encontra tres soluçõoes: (-1, 0), (0, 0), and (2, 0).

Agora que sabemos localizar os pontos criticos, podemos verificar graficamente que (-1, 0) eh um ponto de sela, plotando juntos a superficie e seu plano tangente que é horizontal em z = f(-1, 0) neste ponto. Vamos aproveitar e desenhar todos os planos tangentes nos pontos criticos.

```
> E1:=plot3d(f(x,y), x=-2..3, y=-4..4, style=patch, color=[-y, -y, 1]):
> E2:=plot3d(f(0,0), x=-1..1, y=-2..2, style=patchnogrid, color=red):
> E3:=plot3d(f(-1,0), x=-2..-0, y=-4..4, style=patchnogrid, color=yellow):
> E4:=plot3d(f(2,0), x=1..3, y=-2..2, style=patchnogrid, color=green):
> display({E1, E2, E3,E4}, orientation=[-95, 80]);
```



Passo 2. Usamos o teste da derivada segunda.

Primeiro calculamos as derivadas de ordem dois. O Maple permite fazer isto de várias maneiras:

> **f11:=diff(f1, x);**

$$f11 := \frac{36\,x^2 - 24\,x - 24}{12 + 12\,y^2}$$

> **f12:=diff(f(x,y), x,y); # Equivalently f12:=diff(f1, y) =diff(f2, x).**

$$f12 := -24\,\frac{(12\,x^3 - 12\,x^2 - 24\,x)\,y}{(12 + 12\,y^2)^2}$$

> **f22:=diff(f(x,y), y,y);**

$$f22 := 1152\,\frac{(3\,x^4 - 4\,x^3 - 12\,x^2 + 6)\,y^2}{(12 + 12\,y^2)^3} - 24\,\frac{3\,x^4 - 4\,x^3 - 12\,x^2 + 6}{(12 + 12\,y^2)^2}$$

> **D[2,2](f)(x,y);**

$$1152\,\frac{(3\,x^4 - 4\,x^3 - 12\,x^2 + 6)\,y^2}{(12 + 12\,y^2)^3} - 24\,\frac{3\,x^4 - 4\,x^3 - 12\,x^2 + 6}{(12 + 12\,y^2)^2}$$

> **D[2,2](f);**

$$(x, y) \to 1152\,\frac{(3\,x^4 - 4\,x^3 - 12\,x^2 + 6)\,y^2}{(12 + 12\,y^2)^3} - 24\,\frac{3\,x^4 - 4\,x^3 - 12\,x^2 + 6}{(12 + 12\,y^2)^2}$$

Para calcular a matriz das derivadas segundas, particularmente para uma função de muitas variaveis, é mais rapido usar o comando D[i,j](f) . Assim,

> **DDf:= matrix(2,2, (i, j) -> D[i, j](f));**

$$DDf := \begin{bmatrix} DDf_{1,1} & DDf_{1,2} \\ DDf_{2,1} & DDf_{2,2} \end{bmatrix}$$

Este procedimento automaticamente elimina a entrada de elementos manualmente. É ao mesmo tempo rapido e reduz erros. Nos exemplos 2 e 3 veremos mais a sua utilidade. [Nota de Maple: enquanto o nome D[i, j] significa alguma coisa para o Maple, a nossa escolha para DDf foi arbitraria.]

> **DDf(x,y);**

$$\left[\begin{array}{cc} \dfrac{36\,x^2 - 24\,x - 24}{12 + 12\,y^2} & -24\,\dfrac{(12\,x^3 - 12\,x^2 - 24\,x)\,y}{(12 + 12\,y^2)^2} \\ -24\,\dfrac{(12\,x^3 - 12\,x^2 - 24\,x)\,y}{(12 + 12\,y^2)^2} & 1152\,\dfrac{(3\,x^4 - 4\,x^3 - 12\,x^2 + 6)\,y^2}{(12 + 12\,y^2)^3} - 24\,\dfrac{3\,x^4 - 4\,x^3 - 12\,x^2 +}{(12 + 12\,y^2)^2} \end{array}\right.$$

Se você deseja ver um elemento particular desta matriz, digamos o elemento 2 1, então entre com o comando DDf(x,y)[2,1];

$$-24\,\frac{(12\,x^3 - 12\,x^2 - 24\,x)\,y}{(12 + 12\,y^2)^2}$$

em muitos casos você só quer ver DDf em pontos particulares. Aqui avaliamos esta matriz de segundas derivadas em três pontos criticos.

> **DDf(-1,0);**

$$\begin{bmatrix} 3 & 0 \\ 0 & \frac{-1}{6} \end{bmatrix}$$

Como esta é diagonal e os elementos da diagonal tem sinais opostos, concluimos que é indefinida e assim (-1, 0) é ponto de sela.

> **DDf(0, 0);**

$$\begin{bmatrix} -2 & 0 \\ 0 & -1 \end{bmatrix}$$

Claramente negativa definida (porque?) e assim (0, 0) é um maximo local.

> **DDf(2,0);**

$$\begin{bmatrix} 6 & 0 \\ 0 & \frac{13}{3} \end{bmatrix}$$

Claramente positiva definida e assim (2, 0) é um minimo local.

Exemplo 2. Considere a função $(x^2 - y^2 + 2\,z^2 + 3\,w^2)\,e^{(-x^2 - y^2 - z^2 - w^2)}$

Determine e classifique os pontos criticos da seguinte função de 4 variaveis:

```
> h:=(x,y,z,w) -> (x^2 - y^2 + 2*z^2 +3*w^2)*exp(-x^2 - y^2 -z^2 -w^2);
```

$$h := (x, y, z, w) \to (x^2 - y^2 + 2\,z^2 + 3\,w^2)\,e^{(-x^2 - y^2 - z^2 - w^2)}$$

A função é um exemplo similar ao seguinte exemplo em duas dimensões que pode ser plotado:

```
> plot3d((-y^2 + 3*w^2)*exp(-y^2 -w^2), y=-2.5..2.5, w=-2.5..2.5, style=patch, orientation=
[25,70]);
```



Para o nosso exemplo 4 dimensoes vamos tentar usar o Maple com eficiência máxima --ainda fazendo o possivel para outros como segue.

Passo 1. Encontrar os pontos criticos. Em alguns computadores isto pode levar algum tempo. Podemos encontrar onde o gradiente é zero

```
> grad(h(x,y,z,w), [x,y,z,w]);
```

$$\Big[2\,x\,e^{(-x^2 - y^2 - z^2 - w^2)} - 2\,(x^2 - y^2 + 2\,z^2 + 3\,w^2)\,x\,e^{(-x^2 - y^2 - z^2 - w^2)},$$
$$-2\,y\,e^{(-x^2 - y^2 - z^2 - w^2)} - 2\,(x^2 - y^2 + 2\,z^2 + 3\,w^2)\,y\,e^{(-x^2 - y^2 - z^2 - w^2)},$$
$$4\,z\,e^{(-x^2 - y^2 - z^2 - w^2)} - 2\,(x^2 - y^2 + 2\,z^2 + 3\,w^2)\,z\,e^{(-x^2 - y^2 - z^2 - w^2)},$$
$$6\,w\,e^{(-x^2 - y^2 - z^2 - w^2)} - 2\,(x^2 - y^2 + 2\,z^2 + 3\,w^2)\,w\,e^{(-x^2 - y^2 - z^2 - w^2)}\Big]$$

> convert(", set);# "solve" precisa um conjunto de funcoes, nao # vetor.

$$\{2\,x\,e^{(-x^2-y^2-z^2-w^2)}-2\,(x^2-y^2+2\,z^2+3\,w^2)\,x\,e^{(-x^2-y^2-z^2-w^2)},$$
$$4\,z\,e^{(-x^2-y^2-z^2-w^2)}-2\,(x^2-y^2+2\,z^2+3\,w^2)\,z\,e^{(-x^2-y^2-z^2-w^2)},$$
$$-2\,y\,e^{(-x^2-y^2-z^2-w^2)}-2\,(x^2-y^2+2\,z^2+3\,w^2)\,y\,e^{(-x^2-y^2-z^2-w^2)},$$
$$6\,w\,e^{(-x^2-y^2-z^2-w^2)}-2\,(x^2-y^2+2\,z^2+3\,w^2)\,w\,e^{(-x^2-y^2-z^2-w^2)}\}$$

> solve(");

$$\{y=0,x=0,w=0,z=0\},\{y=0,x=0,w=1,z=0\},\{y=0,x=0,w=-1,z=0\},$$
$$\{x=0,y=1,w=0,z=0\},\{x=0,y=-1,w=0,z=0\},\{y=0,x=0,z=1,w=0\},$$
$$\{y=0,x=0,z=-1,w=0\},\{y=0,x=1,w=0,z=0\},\{y=0,x=-1,w=0,z=0\}$$

ou equivalentemente (e escrevendo um pouco mais), podemos ver onde suas primeiras derivadas são todas nulas:

> solve({diff(h(x,y,z,w), x)=0, diff(h(x,y,z,w), y)=0, diff(h(x,y,z,w), z)=0, diff(h(x,y,z,w), w)=0}, {x,y,z,w});

$$\{y=0,x=0,w=0,z=0\},\{y=0,x=0,w=1,z=0\},\{y=0,x=0,w=-1,z=0\},$$
$$\{x=0,y=1,w=0,z=0\},\{x=0,y=-1,w=0,z=0\},\{y=0,x=0,z=1,w=0\},$$
$$\{y=0,x=0,z=-1,w=0\},\{y=0,x=1,w=0,z=0\},\{y=0,x=-1,w=0,z=0\}$$

Assim existem 9 pontos criticos. Como a função h é simetrica se substituirmos x por -x, o comportamento em (-1,0,0,0) será o mesmo que em (1,0,0,0). Existe uma simetria em todas as variaveis.

Passo 2. O teste da derivada segunda:

> DDh:= matrix(4,4, (i, j) -> D[i, j](h));

$$DDh := \begin{bmatrix} DDh_{1,1} & DDh_{1,2} & DDh_{1,3} & DDh_{1,4} \\ DDh_{2,1} & DDh_{2,2} & DDh_{2,3} & DDh_{2,4} \\ DDh_{3,1} & DDh_{3,2} & DDh_{3,3} & DDh_{3,4} \\ DDh_{4,1} & DDh_{4,2} & DDh_{4,3} & DDh_{4,4} \end{bmatrix}$$

Novamente, isto é o que queremos mas nao é muito informativo. Deveríamos ter terminado o comando com dois pontos e não com ponto e virgula.

> DDh(0,0,0,0);

$$\begin{bmatrix} 2 & 0 & 0 & 0 \\ 0 & -2 & 0 & 0 \\ 0 & 0 & 4 & 0 \\ 0 & 0 & 0 & 6 \end{bmatrix}$$

Esta é indefinida. Assim a origem é um ponto de sela.

> **DDh(1,0,0,0);**

$$\begin{bmatrix} -4\,e^{(-1)} & 0 & 0 & 0 \\ 0 & -4\,e^{(-1)} & 0 & 0 \\ 0 & 0 & 2\,e^{(-1)} & 0 \\ 0 & 0 & 0 & 4\,e^{(-1)} \end{bmatrix}$$

Assim (1,0,0,0) é um ponto de sela. Por simetria (-1,0,0,0) é tambem um ponto de sela.

> **DDh(0,1,0,0);**

$$\begin{bmatrix} 4\,e^{(-1)} & 0 & 0 & 0 \\ 0 & 4\,e^{(-1)} & 0 & 0 \\ 0 & 0 & 6\,e^{(-1)} & 0 \\ 0 & 0 & 0 & 8\,e^{(-1)} \end{bmatrix}$$

asim (0,1,0,0) -- e também (0,-1,0,0) são minimos locais.

> **DDh(0,0,1,0);**

$$\begin{bmatrix} -2\,e^{(-1)} & 0 & 0 & 0 \\ 0 & -6\,e^{(-1)} & 0 & 0 \\ 0 & 0 & -8\,e^{(-1)} & 0 \\ 0 & 0 & 0 & 2\,e^{(-1)} \end{bmatrix}$$

segue que (0,0,1,0) e (0,0,-1,0) são ambos pontos de sela.

> **DDh(0,0,0,1);**

$$\begin{bmatrix} -4\,e^{(-1)} & 0 & 0 & 0 \\ 0 & -8\,e^{(-1)} & 0 & 0 \\ 0 & 0 & -2\,e^{(-1)} & 0 \\ 0 & 0 & 0 & -12\,e^{(-1)} \end{bmatrix}$$

e finalmente (0,0,0,1) e (0,0,0,-1) são ambos maximo local.

Para resumir, h(x,y,z,w) tem dois minimos locais, 2 maximos locais, e 5 pontos de sela. Este calculo poderia ser mais dificil sem o Maple.

Exemplo 3. $1 - 2\,x + 3\,x^2 - x\,y + x\,z - z^2 + 4\,z + y^2 + 2\,y\,z$

Determine e classifique os pontos criticos da sguinte função de tres variaveis.

```
> P:=(x,y,z) -> 1 -2*x +3*x^2 - x*y + x*z - z^2 +4*z + y^2 +2*y*z;
```

$$P := (x, y, z) \to 1 - 2\,x + 3\,x^2 - x\,y + x\,z - z^2 + 4\,z + y^2 + 2\,y\,z$$

Passo 1. Encontre os pontos criticos.

```
> solve({diff(P(x,y,z), x)=0, diff(P(x,y,z), y)=0, diff(P(x,y,z), z)=0}, {x,y,z});
```

$$\{x = 0, y = -1, z = 1\}$$

Assim existe apenas um ponto critico localizado em (0, -1, 1).

Passo 2. Usar o teste da derivada segunda para classificar este ponto critico.Vamos calcular a matriz derivada segunda.

```
> DDP:= matrix(3,3, (i, j) -> D[i, j](P)) ;
```

$$DDP := \begin{bmatrix} 6 & -1 & 1 \\ -1 & 2 & 2 \\ 1 & 2 & -2 \end{bmatrix}$$

Note que P(x,y,z) é um polinomio quadratico, sabiamos que sua matriz de derivadas segundas eh uma matriz constnate. Agora avaliamos a matriz no ponto critico (como acima).

```
> DDP(0, -1, 1);
```

$$\begin{bmatrix} 6 & -1 & 1 \\ -1 & 2 & 2 \\ 1 & 2 & -2 \end{bmatrix}$$

como esta matriz tem ambos elementos positivos e negativos na sua diagonal, sem ir muito longe sabemos que deve ser indefinida, assim o ponto critico eh um ponto de sela.

Observação: se usarmos o Maple para encontrar os autovalores desta matriz poderiamos aplicar o teste dos autovalores --- e entao obter

```
> EV:=eigenvals(DDP(xy,z));
```

$$EV := \frac{1}{3}(-216 + 6\,I\sqrt{6690})^{1/3} + \frac{22}{(-216 + 6\,I\sqrt{6690})^{1/3}} + 2,\ -\frac{1}{6}(-216 + 6\,I\sqrt{6690})^{1/3} - \frac{11}{(-216 + 6\,I\sqrt{6690})^{1/3}} + 2 + \frac{1}{2}I\sqrt{3}\left(\frac{1}{3}(-216 + 6\,I\sqrt{6690})^{1/3} - \frac{22}{(-216 + 6\,I\sqrt{6690})^{1/3}}\right),$$
$$-\frac{1}{6}(-216 + 6\,I\sqrt{6690})^{1/3} - \frac{11}{(-216 + 6\,I\sqrt{6690})^{1/3}} + 2 - \frac{1}{2}I\sqrt{3}\left(\frac{1}{3}(-216 + 6\,I\sqrt{6690})^{1/3} - \frac{22}{(-216 + 6\,I\sqrt{6690})^{1/3}}\right)$$

que tem I (Maple's imaginary unit). Entretanto nós sabemos que os autovalores de uma matriz simétrica são sempre reais. De fato, com algum esforço podemos manipilar a fórmula acima e assim os termos imaginários cancelam. Do gráfico do polinomio caracteristico é evidente que os tres autovalores são reais.

```
> plot(charpoly(DDP(x,y,z), lambda), lambda=-4..8);
```



Para o valor numerico do segundo autovalor, Maple dá

> **evalf(EV[2]);**

$$-3.018768109 - .2\ 10^{-9}\ I$$

que é incorreto, tem uma parte imaginaria muito pequena. Moral da historia: não aceite de olhos fechados qualquer coisa. Todas os cálculos numéricos tem erros, o Maple usa 10 digitos significativos como defaut. Se queremos o valor numérico com 20 digitos, escrevemos

> **evalf(EV[2], 20);**

$$-3.0187681086892283920 - .2\ 10^{-19}\ I$$

Esta é uma evidencia numérica do fato que a parte imaginária é zero. Por esta razão vamos testar se a matriz é positiva definida. Por razões como esta é que o teste do determinante para verificar se uma matriz é positiva definida é preferivel em algumas situações. Entretanto, para uma matriz 20x20 os cálculos para os testes são muito complicados a menos que a matriz seja muito especial, o que ocorre frequentemente em aplicações interessantes.

P é um polinomio quadratico e assim podemos escrevê-lo como P(x, y, z) = P(X) = <X, A X> + <b, X> + c, onde

> **X:= vector([x, y, z]);**

$$X := [x, y, z]$$

c = P(0, 0, 0) e b é o gradiente de P na origem:

> c:=P(0, 0, 0);

$$c := 1$$

> b:=subs(x=0, y=0, z=0, grad(P(x,y,z), [x,y,z]));

$$b := [-2, 0, 4]$$

A matriz A é exatamente 1/2 da matriz das derivadas segundas

> A:=evalm((1/2)*DDP);

$$A := \begin{bmatrix} 3 & \frac{-1}{2} & \frac{1}{2} \\ \frac{-1}{2} & 1 & 1 \\ \frac{1}{2} & 1 & -1 \end{bmatrix}$$

Como

> det(A);

$$\frac{-13}{2}$$

det A é não zro então A é invertivel. Assim sabemos que podemos transladar X --> X - r então P tem uma forma simples P(X) = <(X-r), A (X-r)> + c1, onde r é um vetor a ser encontrado e c1 um escalar. Então a reta

X = r é o eixo deste polinomio quadratico. Em coordenadas transladadas V = X - r, o polinomio quadratico P tem a forma mais simples P = <V, A V> + c1 .

Para encontrar r podemos usar a formula dada,

> r := evalm(-(1/2)*inverse(A)&*b);

$$r := [0, -1, 1]$$

ou então calcula-lo diretamente. Agora vamos calcula-lo diretamente usando a observação que o eixo para um polinomio quadratico não degenerado, isto é, com A invertivel, é sempre o ponto critico de P.

Nós calculamo isto no Passo 1 acima.

De qualquer modo encontramos r = (0, -1, 1). Também, a constante c1 é exatamente o valor de P no ponto critico X = r:

> c1:=P(0, -1, 1);

$$c1 := 3$$

Finalmente, verificamos que <X-r, A(X-r)> + c1 é exatamente nosso polinomio:

```
> innerprod(X-r, A, X-r) + c1;
```

$$3x^2 - xy - 2x + xz + y^2 + 2zy + 4z + 1 - z^2$$

```
> P(x,y,z) - ";
```

$$0$$

```
> r:='r': x:='x': y:='y': z:='z': w:='w': # para limpar
```

Exemplo 4. Discuta os pontos criticos da função

```
> g:= (x,y) -> x^2 + c*(x - y)^4;
```

$$g := (x, y) \to x^2 + c\,(x - y)^4$$

para varios valores da constante c.

Passo 1. Encontrar os pontos criticos.

```
> g1:=diff(g(x,y), x);g2:=diff(g(x,y), y);#nao essencial,mas as vezes #util
```

$$g1 := 2x + 4\,(x - y)^3$$

$$g2 := -4\,(x - y)^3$$

```
> solve( { diff(g(x,y), x)=0, diff(g(x,y),y)=0}, {x,y});
```

$$\{x = 0, y = 0\}$$

Maple achou apenas uma solucao (0, 0). Esta solução nao é completa porque Maple assumiu c não zero. Se c = 0, então todos os pontos (0, y) sobre o eixo y são pontos criticos.

Passo 2. O teste da derivada segunda.

```
> DDg:= matrix(2,2, (i, j) -> D[i, j](g));
```

$$DDg := \begin{bmatrix} DDg_{1,1} & DDg_{1,2} \\ DDg_{2,1} & DDg_{2,2} \end{bmatrix}$$

No ponto critico (0, 0), temos

```
> DDg(0, 0);
```

$$\begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}$$

Esta matriz é não invertivel assim o teste da derivada segunda nos dá informação insuficiente. Afirmamos que a origem pode ser é um minimo local, ou um ponto de sela , mas não um ponto de maximo local, pois nenhum dos autovalores sao positivos.

Para analise deste exemplo podemos fazer casos basicos. Nesta situação especial se c > 0 vemos da formula explicita de g(x,y) que a origem eh um minimo estrito local, enquanto se c < 0 a origem eh um ponto de sela. Finalmente, se c = 0, entao TODOS os pontos da reta y = x sao minimos locais, mas não minimos estritos locais.

Vamos ver os gráficos nos casos especiais c=1, c=0, e c=-1. Primeiro o caso c=1:

```
> plot3d(x^2 + (x-y)^4, x=-3..3, y=-3..3, view=-2..20, style=patch, orientation=[15,75], axes=normal);
```



A seguir c = -1 (ponto de sela)

```
> plot3d(x^2 - (x-y)^4, x=-2..2, y=-2..2, view=-2..20,style=patch, orientation=[15,75], axes=normal);
```



e c = 0 (minimo local sobre a reta y = x, mas não estrito).

\> **plot3d(x^2 , x=-2..2, y=-2..2, style=patch, orientation=[15,75], axes=normal);**



Exemplo 5. (ponto critico degenerado) Discuta os pontos criticos de

$$(x^2 + (y - 2)^2)\,(x^2 - (y + 2)^2)$$

> g:=(x,y) -> (x^2 + (y-2)^2)*(x^2 - (y+2)^2);

$$g := (x, y) \to (x^2 + (y-2)^2)\,(x^2 - (y+2)^2)$$

Passo 0. Como esta é uma superficie bidimensional, podemos tirar algumas informações do grafico.

> plot3d(g(x,y), x=-10..10, y=-10..10, style=patch, orientation=[35, 80], axes=normal);



Nesta escala, tudo se passa como $x^4 - y^4$ **, por causa dos coeficientes lideres. Assim vamos tentar uma escala menor**

> plot3d(g(x,y), x=-2..2, y=-3..3, style=patch, orientation=[35, 80]);



Ainda não está claro, embora podemos ver que existem um máximo local e um ponto de sela. Uma escala menor ainda para x.

```
> plot3d( g(x,y), x=-1..1, y=-3..3, style=patch, orientation=[35, 80]);
```



Agora vemos que existem um máximo local e dois pontos de sela-- talvez. Um cálculo poderia ajudar.

Passo 1. Encontre os pontos criticos. Calculamos o gradiente e encontramos onde é zero.

> **grad(g(x,y), [x,y]);**

$$[2x(x^2-(y+2)^2)+2(x^2+(y-2)^2)x, (2y-4)(x^2-(y+2)^2)+(x^2+(y-2)^2)(-2y$$

> **convert(",set);**

> **solve(",{x,y});**

$$\{(2y-4)(x^2-(y+2)^2)+(x^2+(y-2)^2)(-2y-4), 2x(x^2-(y+2)^2)+2(x^2+(y-2$$

$$\{x=0, y=0\}, \{x=0, y=2\}, \{x=0, y=-2\},$$
$$\{x=2\,\mathrm{RootOf}(\_Z^2-2\_Z+2), y=-2+2\,\mathrm{RootOf}(\_Z^2-2\_Z+2)\},$$
$$\{y=-2-2\,\mathrm{RootOf}(\_Z^2+2\_Z+2), x=2\,\mathrm{RootOf}(\_Z^2+2\_Z+2)\}$$

Assim existem três pontos criticos reais: (0, 0), (0, 2), and (0, -2). [As raizes de z^2 + 4z + 8=0 e z^2 - 4z +8 = 0 sao complexas].

Passo 2. Teste da derivada segunda. Primeiro calculamos a matriz das derivadas segundas.

> **DDg:=matrix(2,2, (i,j) -> D[i,j](g));**

$$DDg := \begin{bmatrix} DDg_{1,1} & DDg_{1,2} \\ DDg_{2,1} & DDg_{2,2} \end{bmatrix}$$

Em (0, 2):

> **DDg(0,2);**

$$\begin{bmatrix} -32 & 0 \\ 0 & -32 \end{bmatrix}$$

Claramente um máximo local.

Em (0, -2):

> **DDg(0, -2);**

$$\begin{bmatrix} 32 & 0 \\ 0 & -32 \end{bmatrix}$$

e igualmente claramente sela.

Em (0, 0):

> **DDg(0, 0);**

$$\begin{bmatrix} 0 & 0 \\ 0 & 16 \end{bmatrix}$$

Esta matrix de derivada segunda é não invertivel e assim este ponto critico é degenerado. Não poderia ser um maximo local (porque?), mas pode ser ou um minimo local ou um ponto de sela. Para decidir devemos examinar o comportamento na origem mais cuidadosamente. Podemos usar um calculo grafico ou um enfoque computacional. Neste caso, usaremos informacao adicional.

ENFOQUE COMPUTACIONAL:

```
> expand(g(x,y));
```

$$x^4 - 8x^2y - y^4 + 8y^2 - 16$$

perto da origem $8y^2$ **é muito maior que** y^4 **. Como g(0,0) = -16,**

g(x,y) - g(0,0) ~ x^4 - 8yx^2 + 8y^2. Completando o quadrado isto é g(x,y) - g(0,0) ~ (x^2 - 4y)^2 - 8y^2.

Desta formula, é claro que existem pontos perto da origem onde o lado direito é ambos positivo e negativo. Assim a origem eh um ponto de sela.

Podemos fatorar

(x^2 -4y)^2 -8y^2 = [(x^2 - 4y) - sqrt(8) y] [(x^2 - 4y) + sqrty(8) y] = (x^2 -ay)(x^2 -by)

onde a = 4 + sqrt(8) ~ 6.8 e b = 4 - sqrt(8) ~ 1.2. Assim, perto da origem g(x,y) - g(0,0) ~ 0 sobre duas parabolas

y = (1/a)x^2 e y = (1/b)x^2, é negativo entre estas parabólas, e positiva ambos na região acima e abaixo de ambas as parabólas. Podemos ver neste segundo grafico.

Nesta analise, temos uma tipica situação onde o Maple foi capaz de resolver, mas devemos ainda fazer algum detalhe. Este é um dos pontos no uso de um programa como o Maple. Vamos calcular as partes standard assim você pode concentrar sua energia em aspectos mais adequados.

ENFOQUE GRÁFICO:

Usamos o mesmo plot que acima, mas restringimos nossa atenção perto da origem -- e assim plotar o plano tangente perto da origem -- e também plotamos o plano tangente no ponto critico. Como g(0,0) = -16, a equação do plano tangente é exatamente z = -16.

```
> with(plots):
G1:=plot3d( g(x,y), x=-0.7..0.7, y=-0.3..0.5, style=patch, color=[.4, .4, 1]):
```

```
> G2:=plot3d( g(0,0), x=-0.7..0.7, y=-0.3..0.5, style=patchnogrid, color=blue):
```

> display({G1, G2}, orientation=[30,80]);



Como o plano tangente na origem corta a superficie, o ponto critico é de sela.

No próximo gráfico, você pode ver o significado das duas parabolas encontradas acima.

> G3:=plot3d(g(x,y), x=-0.7..0.7, y=-0.3..0.5, grid=[50,50],style=patch, color=[.4, .4, 1]):

> display({G3, G2}, orientation=[15,20]);



>

>